' InsertBibEntry


' ============


Sub MAIN


On Error Goto Bye


tag$ = InputBox$("Please type the entry name: ", "Reference Entry")


InsertFieldChars


Insert "SET " + tag$ + " "


InsertFieldChars


Insert "SEQ Reference"


CharRight 4


Insert "["


InsertFieldChars


Insert "REF " + tag$


CharRight 2


Insert "]"


EditBookmark .Name = tag$, .Add


CharLeft 1


StartOfLine 1


UpdateFields


EndOfLine


Bye:


End Sub








' InsertBibEntry


' ============


Sub MAIN


On Error Goto Bye


tag$ = InputBox$("Please type the entry name: ", "Reference Entry")


InsertFieldChars


Insert "SET " + tag$ + " "


InsertFieldChars


Insert "SEQ Reference"


CharRight 4


Insert "["


InsertFieldChars


Insert "REF " + tag$


CharRight 2


Insert "]"


EditBookmark .Name = tag$, .Add


CharLeft 1


StartOfLine 1


UpdateFields


EndOfLine


Bye:


End Sub








'ResolveRefs


'==========


Declare Function GetActiveWindow Lib "User" As Integer	'used to turn screen updates to OFF


Declare Function LockWindowUpdate Lib "User"(hWnd As Integer) As Integer





Dim Shared Refs$(256)					' Array where references tags are stored.


'When filled, there will be NumOfRefs entries, indexed by RefIndex





Sub Main





Begin Dialog UserDialog 334, 190, "Microsoft Word"


	Text 10, 6, 117, 13, "Reference File:"


	TextBox 10, 20, 310, 18, .RefFile


	OKButton 24, 158, 88, 21


	CancelButton 207, 157, 88, 21


	Text 27, 126, 229, 13, "Progress shown on status bar."


	Text 26, 141, 171, 13, "This may take a while."


	GroupBox 26, 46, 248, 55, "When done", .WhenDone


	CheckBox 36, 60, 152, 16, "&Update all fields", .UpdateFields


	CheckBox 36, 78, 227, 16, "&Close reference document", .CloseReferDoc


End Dialog





Dim Dlg As UserDialog


Dlg.RefFile = "c:\thesis\refer.doc"


If Not Dialog(Dlg) Then Goto DoneResolveRefs		'User pressed cancel button


RefFile$ = Dlg.RefFile					'store user's refer filname 


UpdateFieldsFlag = Dlg.UpdateFields


CloseRefDocFlag = Dlg.CloseReferDoc


'RefFile$ = "c:\ido\art\second\docs\optlet\refer.doc"


EditBookmark .Name = "position", .Add	'Save current cursor position


' Delete the old bibliography


If Not ExistingBookmark("References") Then 		'If there is no 'reference' bookmark, 


	MsgBox "No References Bookmark", "Error", 48	'display a message and abort macro


	Goto DoneResolveRefs


End If


Print "Deleting the old reference list."


Echo(0)							'turn off screen updates


If ExistingBookmark("RefEnd") Then			'If a bookmark exist than


	EditGoTo .Destination = "References"		'go to it and select all the


	ExtendSelection					'references till the end of them,


	EditGoTo .Destination = "RefEnd"			'marked by bookmark 'refend'


	Cancel						'zap them,


	EditClear					'and delete the bookmark 'refend'


	EditBookmark "RefEnd", .Delete


End If





' Make a list of unique references Tags in document


Echo(1)


Print "Searching and collecting Reference Tags."


Echo(0)


StartOfDocument





'Backup document view states


Dim dlg1 As ToolsOptionsView


GetCurValues dlg1


ShowAllState = dlg1.ShowAll


ViewFieldCodesStatus = Dlg1.FieldCodes


HiddenStatus = dlg1.Hidden





'set states needed for macro


dlg1.ShowAll = 1


dlg1.FieldCodes = 1


dlg1.Hidden = 1


ToolsOptionsView dlg1





NumOfRefs = 0


EditFind .Find = "^d ref ", .Direction = 0, .MatchCase = 1, .Format = 0


		RefTagText$ = Mid$(Selection$(), 7, Len(Selection$()) - 8)


		If Not IsIn(RefTagText$, NumOfRefs) Then	'If this is a new (unique) reference 


			NumOfRefs = NumOfRefs + 1	'then add it to the reference tag list


			Refs$(NumOfRefs) = RefTagText$


		End If


While EditFindFound()


	EditFind


		RefTagText$ = Mid$(Selection$(), 7, Len(Selection$()) - 8)


		If Not IsIn(RefTagText$, NumOfRefs) Then	'If this is a new (unique) reference 


			NumOfRefs = NumOfRefs + 1	'then add it to the reference tag list


			Refs$(NumOfRefs) = RefTagText$


		End If


Wend


DoneRefs:					


WinDoc = Window()				'save window  number


' Open the reference database, and place the selected references on the spike.


Echo(1)


Print "Searching and collecting reference entries from file: " + RefFile$ + " ."


Echo(0)


FileOpen .Name = RefFile$





For Counter = 1 To NumOfRefs


	If ExistingBookmark(Refs$(Counter)) Then		'copy the reference entry


		EditGoTo .Destination = Refs$(Counter)	'onto the spike


		ParaDown 1, 1


		EditCopy


		Spike


		EditPaste


	End If


Next Counter


If CloseRefDocFlag Then


	FileClose 2


End If





' Return to the document, and insert the references


WindowList winDoc				'return to correct window number


Echo(1)


Print "Inserting reference list and updating fields."


Echo(0)


EditGoTo .Destination = "References"


InsertSpike


EditBookmark "RefEnd", .Add


ExtendSelection


EditGoTo .Destination = "References"


UpdateFields





If UpdateFieldsFlag Then


	StartOfLine


	ExtendSelection


	StartOfDocument


	UpdateFields


End If





DoneResolveRefs:


Echo(1)


EditGoTo .Destination = "position"


EditBookmark .Name = "position", .Delete


Print "Done Resolve References."





'Restore Document states


Dim dlg2 As ToolsOptionsView


GetCurValues dlg2


dlg2.ShowAll = ShowAllState


Dlg2.FieldCodes = ViewFieldCodesStatus


dlg2.Hidden = HiddenStatus


ToolsOptionsView dlg2





End Sub





Function IsIn(r$, n)				' Returns 1 if r$ is in n references collected so far.


	For Counter = 1 To n


		If Refs$(Counter) = r$ Then


			IsIn = - 1


			Goto DoneIsIn


		End If


	Next Counter


	IsIn = 0


DoneIsIn:


End Function





Sub Echo(n)





'This version of Echo uses LockWindowUpdate, available only in Windows 3.1


	Select Case n


		Case 0		' OFF


			CurrWindow = GetActiveWindow


			result = LockWindowUpdate(CurrWindow)


		Case Else	' ON


			result = LockWindowUpdate(0)


	End Select


Bye:


End Sub


--------------------------------------------


